QUICK REFERENCE · TK18

Hall Effect Sensor

A magnet near the chip reads LOW. Either pole.

VCC sets the signal With no magnet near, the chip drives SIGNAL up to VCC. **On an ESP32, ESP32-S3 or Pico, feed VCC from 3V3, never 5V**: 5 V would sit on a 3.3 V input for as long as the magnet is away.

Pins, and which to wire

Parts up, header at the bottom. Wire GND, VCC and SIGNAL; NC is connected to nothing on the board.

CC6201ST Hall sensor
Hold the magnet here, flat face on. Either pole.

10 kΩ pull-up to VCC
The chip drives SIGNAL both ways. Use INPUT.

GND is the square pad
Count from it. The back prints no pin names.



Printed magnet
Artwork only: nothing under it senses a field.

Red LED, 1 kΩ
Lights while SIGNAL is LOW: lit means a magnet.

NC
Connected to nothing on the board.

Front	GND	VCC	NC	SIGNAL
Read it	GND	VCC	●	SIGNAL

LOW = magnet The chip is active low and omnipolar: either pole past about 40 gauss pulls SIGNAL LOW, and it lets go below 32. It does not latch.

How close a magnet must come

5 × 2 mm disc, N35

11 mm from the chip, at most

Switches it inside about 11 mm and lets go beyond about 12 mm.

10 × 3 mm disc, N35

20 mm from the chip, at most

Switches it inside about 20 mm and lets go beyond about 22 mm.

20 × 5 mm disc, N35

39 mm from the chip, at most

Switches it inside about 39 mm and lets go beyond about 42 mm.

The field falls away steeply with distance, so the magnet has to come closer than most people expect. A flat fridge magnet may have to touch the board. No magnet is included.

Calculated from the on-axis field of an N35 neodymium disc (about 1.2 T), face on and on axis, at the chip's typical 40 G and 32 G. Not measured: test your own gap and build well inside it.

A door left open

Arduino IDE · Board: your board · ESP32-S3: USB CDC On Boot: Enabled

```
// SIGNAL's pin. Uno: 2. ESP32: 25. ESP32-S3: 4. Pico: 15.
const int HALL_PIN = 4;
const int LED_PIN = 5;     // the alarm: a TK01 or any LED
const unsigned long ALARM_AFTER_MS = 10000;

bool wasShut;
unsigned long openedAt;

void setup() {
  Serial.begin(115200);
  pinMode(HALL_PIN, INPUT);   // no pull needed
  pinMode(LED_PIN, OUTPUT);
  wasShut = digitalRead(HALL_PIN) == LOW;
  openedAt = millis();
}

void loop() {
  bool shut = digitalRead(HALL_PIN) == LOW;  // magnet near
  if (!shut && wasShut) {     // just opened: start the clock
    openedAt = millis();
    Serial.println("opened");
  }
  wasShut = shut;
  bool alarm = !shut && millis() - openedAt >= ALARM_AFTER_MS;
  digitalWrite(LED_PIN, alarm ? HIGH : LOW);
}
```

```
opened
opened
opened
```

Serial Monitor at 115200: a line per opening. The LED lights once the door has stood open 10 s.

Wiring

	FRONT	UNO	ESP32	S3	PICO
1	**GND**	GND	GND	GND	GND
2	**VCC**	5V	3V3	3V3	3V3
3	**NC**				
4	**SIGNAL**	D2	GPIO 25	GPIO 4	GP15

Hole 1 is the square pad. The alarm LED goes on a second pin: D9 (Uno), GPIO 4 (ESP32), GPIO 5 (S3), GP14 (Pico).

Once every 22 ms

The chip sleeps between looks at the field, about one every 22 ms, and holds SIGNAL in between. So SIGNAL can change up to 22 ms late, and a magnet in range for less than that can pass unseen. Fine for doors, lids and slow wheels: one magnet on a wheel is counted reliably only well below about 1400 rpm, and far lower when the magnet is in range for a small part of each turn.

When it does not work

A magnet does nothing

Hold it over the small chip at the top left, flat face on and close: within about a centimetre for a small disc.

LED lit, sketch sees nothing

The chip saw the magnet. Check the pin number in the sketch, the SIGNAL wire and a shared GND.

Reads a magnet, LED lit

Something magnetic is near the chip: a speaker, a tool, a phone case. Move the block.

It reads backwards

LOW is the magnet on this block. A sketch that treats HIGH as the magnet has it the wrong way round.

Reads a magnet, LED dark

VCC is probably missing, and SIGNAL is drifting. Count from the square pad: GND, VCC, NC, SIGNAL.

It misses a fast magnet

The chip looks about every 22 ms. Slow the wheel or keep the magnet in range longer.

Specifications

Sensor	CrossChip CC6201ST, omnipolar Hall switch	**Output**	Active low: a magnet reads LOW
Switches	LOW at 40 G, back HIGH below 32 G	**Response**	One look every 22 ms (typical)
Indicator	Red LED, lit on LOW, about 3.0 mA at 5 V	**Supply**	3.3 V or 5 V, the same as your board
Current	About 5 µA; 3.5 mA with the LED lit	**Size**	22.4 × 30.4 mm, 4-pin right-angle header